G01ERF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G01ERF returns the probability associated with the lower tail of the von Mises distribution between $-\pi$ and π through the function name.

2 Specification

```
real FUNCTION G01ERF(T, VK, IFAIL)
INTEGER              IFAIL
real                 T, VK
```

3 Description

The von Mises distribution is a symmetric distribution used in the analysis of circular data. The lower tail area of this distribution on the circle with mean direction $\mu_0 = 0$ and concentration parameter kappa, κ, can be written as:

$$\Pr(\Theta \leq \theta : \kappa) = \frac{1}{2\pi I_0(\kappa)} \int_{-\pi}^{\theta} e^{\kappa \cos \Theta} d\Theta,$$

where θ is reduced modulo 2π so that $-\pi \leq \theta < \pi$ and $\kappa \geq 0$. Note that if $\theta = \pi$ then this routine returns a probability of 1. For very small κ the distribution is almost the uniform distribution, where as for $\kappa \to \infty$ all the probability is concentrated at one point.

The method of calculation for small κ involves backwards recursion through a series expansion in terms of modified Bessel functions, while for large κ an asymptotic Normal approximation is used.

In the case of small κ the series expansion of $\Pr(\Theta \leq \theta\colon \kappa)$ can be expressed as

$$\Pr(\Theta \leq \theta : \kappa) = \frac{1}{2} + \frac{\theta}{(2\pi)} + \frac{1}{\pi I_0(\kappa)} \sum_{n=1}^{\infty} n^{-1} I_n(\kappa) \sin n\theta,$$

where $I_n(\kappa)$ is the modified Bessel function. This series expansion can be represented as a nested expression of terms involving the modified Bessel function ratio R_n,

$$R_n(\kappa) = \frac{I_n(\kappa)}{I_{n-1}(\kappa)}, \qquad n = 1, 2, 3, \ldots$$

which is calculated using backwards recursion.

For large values of κ (see Section 7) an asymptotic Normal approximation is used. The angle Θ is transformed to the nearly Normally distributed variate Z,

$$Z = b(\kappa) \sin \frac{\Theta}{2},$$

where

$$b(\kappa) = \frac{\sqrt{\frac{2}{\pi}} e^{\kappa}}{I_0(\kappa)}$$

and $b(\kappa)$ is computed from a continued fraction approximation. An approximation to order κ^{-4} of the asymptotic normalizing series for z is then used. Finally the Normal probability integral is evaluated.

For a more detailed analysis of the methods used see Hill [2].

4 References

[1] Mardia K V (1972) *Statistics of Directional Data* Academic Press

[2] Hill G W (1977) Algorithm 518: Incomplete Bessel function I_0: The von Mises distribution *ACM Trans. Math. Software* **3** 279–284

5 Parameters

1: T — ***real*** *Input*

On entry: the observed von Mises statistic, θ, measured in radians.

2: VK — ***real*** *Input*

On entry: the concentration parameter κ, of the von Mises distribution.

Constraint: VK ≥ 0.

3: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

On entry, VK < 0 and G01ERF returns 0.

7 Accuracy

The routine uses one of two sets of constants depending on the value of ***machine precision***. One set gives an accuracy of six digits and uses the Normal approximation when VK ≥ 6.5, the other gives an accuracy of 12 digits and uses the Normal approximation when VK ≥ 50.

8 Further Comments

Using the series expansion for small κ the time taken by the routine increases linearly with κ; for larger κ, for which the asymptotic Normal approximation is used, the time taken is much less.

If angles outside the region $-\pi \leq \theta < \pi$ are used care has to be taken in evaluating the probability of being in a region $\theta_1 \leq \theta \leq \theta_2$ if the region contains an odd multiple of π, $(2n+1)\pi$. The value of $F(\theta_2;\kappa) - F(\theta_1;\kappa)$ will be negative and the correct probability should then be obtained by adding one to the value.

9 Example

Four values from the von Mises distribution along with the values of the parameter κ are input and the probabilities computed and printed.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G01ERF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             P, T, VK
      INTEGER          I, IFAIL, N
*     .. External Functions ..
      real             G01ERF
      EXTERNAL         G01ERF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01ERF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      DO 20 I = 1, N
         READ (NIN,*) T, VK
         IFAIL = 1
         P = G01ERF(T,VK,IFAIL)
         WRITE (NOUT,99999) P
   20 CONTINUE
      STOP
*
99999 FORMAT (' P = ',F10.4)
      END
```

9.2 Program Data

```
G01ERF Example Program Data
 4
 7.0  0.0
 2.8  2.4
 1.0  1.0
-1.4  1.3
```

9.3 Program Results

```
 G01ERF Example Program Results

 P =     0.6141
 P =     0.9983
 P =     0.7944
 P =     0.1016
```